
LC_transLine_LRC
*
*           RIN         1     2    3     4      5     6    7     8     9     10
*   VIN                   ___   ___   ___   ___   ___   ___   ___   ___   ___   ___    ___
*       ___/\  /\  /\____|LC1|_|LC1|_|LC1|_|LC1|_|LC1|_|LC2|_|LC2|_|LC2|_|LC2|_|LC2|__|OUT|
*     _|__   \/  \/  _|_ |___| |___| |___| |___| |___| |___| |___| |___| |___| |___|  |___|
*    /_   \          ___  _|_   _|_   _|_   _|_   _|_   _|_   _|_   _|_   _|_   _|_     |__
*   // \   \    CIN   |   ///   ///   ///   ///   ///   ///   ///   ///   ///   ///     | _|_
*   \   \_//         _|_                                    _  _  _                     | ___ COUT
*    \____/          ///                                   / \/ \/ \ VLG                |  |
*     _|_                                                __| () () |_____/\  /\  /\_____|  |
*     ///                                               _|_   LOUT         \/  \/ ROUT1    |
*                                                       ///                                |VCG
*                                                                        _/\  /\  /\_______|
*                                                                      _|_  \/  \/  ROUT2
*           www.idea2ic.com dsauersanjose@aol.com  4/15/08             ///
*                                                                      

.OPTIONS  GMIN=1e-18    METHOD=euler   ABSTOL=1e-18  TEMP=27 srcsteps = 1  gminsteps = 1    
VIN       VIN    0      PWL( 0 0 2n 0 3.0n 1  18n 1 19.0n 0)   
RIN       VIN    1      50
C1        VIN    0      -1.6p
XLC1      1      2      LC1
XLC2      2      3      LC1
XLC3      3      4      LC1 
XLC4      4      5      LC1 
XLC5      5      6      LC1 
XLC6      6      7      LC1
XLC7      7      8      LC1
XLC8      8      9      LC1
XLC9      9      10     LC1
XLC10     10     OUT    LC1
C2        OUT    0      -1.6p

ROUT1     OUT    VLG    50
LOUT      VLG    0      1p
COUT      OUT    VCG    30f
ROUT2     VCG    0      5k

.tran     0.01n   30n 0 30n

*#0===The_Goal_Play_With_Reflections=====
.control
*#1===First_50_Ohms=======================
run
plot          v(1) v(6) out   title                 R1_R2_L_C_50_5K_1p_8f
*#2===Next_80_Ohms=======================
alter        ROUT1    resistance  =    80
run
plot          v(1) v(6) out   title                 R1_R2_L_C_80_5K_80n_8f
*#3===Next_20_Ohms=======================
alter        ROUT1    resistance  =    20
run
plot          v(1) v(6) out   title                 R1_R2_L_C_20_5K_80n_8f
*#4===Next_80nH_series_50_Ohms==========
alter        ROUT1    resistance  =    50
alter        LOUT    inductance   =    80n
run
plot             v(1) v(6) out   title              R1_R2_L_C_50_5K_80n_8f
*#4===Next_30pF_Shunt_50_Ohms===========
alter        LOUT    inductance   =    1p
alter        COUT    capacitance  =    30p
alter        ROUT2    resistance  =    1m
run
plot             v(1) v(6) out   title              R1_R2_L_C_50_1m_1p_30p
*#4===Next_30pF_Shunt_50_Ohms===========
alter        LOUT    inductance   =    80n
alter        COUT    capacitance  =    30u
alter        ROUT1    resistance  =    1m
alter        ROUT2    resistance  =    50
run
plot             v(1) v(6) out   title               R1_R2_L_C_1m_50_80n_30u



*#4===Next_30pF_series_50_Ohms==========
alter        LOUT    inductance   =    1p
alter        COUT    capacitance  =    30p
alter        ROUT1    resistance  =    10k
alter        ROUT2    resistance  =    50
run
plot             v(1) v(6) out   title              R1_R2_L_C_10k_50_1p_30p

.endc

.SUBCKT LC1  IN    OUT
L1      IN   OUT   8n
C1      OUT  0     3.2p
.ENDS   LC1

*                               .SUBCKT LC1 IN OUT
*          _  _  _  L1
*    ___  / \/ \/ \    ___        ___    _______     ___
*   |IN |_| () () |___|OUT|      |IN |__|       |___|OUT|
*   |___|           | |___|      |___|  |  LC1  |   |___|
*                  _|_                  |_______|
*                  ___ C1                  _|_
*                   |                      ///
*                  _|_
*                  ///    8nH_3.2pF =>50Ohms @ 1Ghz  50*3.2p =160ps
*
*    http://www.idea2ic.com/PlayWithJavascript/L_C_R_F.html
*

.END

=====================END_OF_SPICE============================

*#0===The_Goal_Play_With_Reflections=====

Adding all the various output impedances 
to the end of a transmission line 
appear to produce all the results 
describe in the HP app note..

Time Domain Reflectometry  Theory
Application Note 1304-2

All missmatches have their own characteristics
Using a reflection method makes sense.

*#1===First_50_Ohms=======================
run
plot          v(1) v(6) out   title                 R1_R2_L_C_50_5K_1p_8f
Ideal 50 Ohms



*#2===Next_80_Ohms=======================
alter        ROUT1    resistance  =    80
run
plot          v(1) v(6) out   title                 R1_R2_L_C_80_5K_80n_8f

80 Ohms

*#3===Next_20_Ohms=======================
alter        ROUT1    resistance  =    20
run
plot          v(1) v(6) out   title                 R1_R2_L_C_20_5K_80n_8f

20 Ohms



*#4===Next_80nH_series_50_Ohms==========
alter        ROUT1    resistance  =    50
alter        LOUT    inductance   =    80n
run
plot             v(1) v(6) out   title              R1_R2_L_C_50_5K_80n_8f
50 Ohms + 80nH in series

*#4===Next_30pF_Shunt_50_Ohms===========
alter        LOUT    inductance   =    1p
alter        COUT    capacitance  =    30p
alter        ROUT2    resistance  =    1m
run
plot             v(1) v(6) out   title              R1_R2_L_C_50_1m_1p_30p

50 Ohms Shunt by 30pF



*#4===Next_30pF_Shunt_50_Ohms===========
alter        LOUT    inductance   =    80n
alter        COUT    capacitance  =    30u
alter        ROUT1    resistance  =    1m
alter        ROUT2    resistance  =    50
run
plot             v(1) v(6) out   title               R1_R2_L_C_1m_50_80n_30u

50 Ohms shunt by 80nH

*#4===Next_30pF_series_50_Ohms==========
alter        LOUT    inductance   =    1p
alter        COUT    capacitance  =    30p
alter        ROUT1    resistance  =    10k
alter        ROUT2    resistance  =    50
run
plot             v(1) v(6) out   title              R1_R2_L_C_10k_50_1p_30p

50 Ohms + 30pF in series




